```
> ### All solutions of paper's congruences (2.5) and (2.6) to 10^6.mws
```

In this worksheet we exhibit (purely for demonstration purposes) all solutions to 10^6 of:

- $\left(\frac{1}{3}\{n-1\}\right)_n! = 1 \pmod{n}$ for $n = p^{\alpha} q_1^{\beta_1} \ldots q_s^{\beta_s}$, prime $p = 1 \pmod 3$ and **distinct** primes $q_1, \ldots, q_s = -1, \ldots, -1 \pmod 3$
- $\left(\frac{1}{6}\{n-1\}\right)_n! = 1 \pmod{n}$ for $n = p^{\alpha} q_1^{\beta_1} \ldots q_s^{\beta_s}$, prime $p = 1 \pmod 6$ and **distinct** primes $q_1, \ldots, q_s = -1, \ldots, -1 \pmod 6$

These are the (2.5) and (2.6) of our paper.

In both cases (consistent with the notation of our paper *A role for generalised Fermat numbers*) we set $w := q_1^{\beta_1} \ldots q_s^{\beta_s}$.

Note. It should be observed that for those cases where $n = -1 \pmod 3$ or $n = -1 \pmod 6$ then $\left(\frac{1}{3}\{n-1\}\right)_n!$ and $\left(\frac{1}{6}\{n-1\}\right)_n!$ are $\mathrm{floor}\left(\frac{1}{3}\{n-1\}\right)_n!$ and $\mathrm{floor}\left(\frac{1}{6}\{n-1\}\right)_n!$ respectively.

Some comments concerning the upcoming Procedures section.

1. The first one - PI (which we use with M = 3 or 6 and i = 1) - provide brute force computations of the Gauss factorials

$$\left(\frac{1}{3}\{n-1\}\right)_n! \text{ and } \left(\frac{1}{6}\{n-1\}\right)_n!$$

2. PHI3 and PHI6 are the computations of the D. H. Lehmer $\phi(3,1,w)$ and $\phi(6,1,w)$ of our paper

   Those $\phi(3,1,w)$ and $\phi(6,1,w)$ are the number of residues in the intervals $\left[1, \mathrm{floor}\left(\frac{1}{3}\{w-1\}\right)\right]$ and $\left[1, \mathrm{floor}\left(\frac{1}{6}\{w-1\}\right)\right]$. that are relatively prime to w. The D. H. Lehmer formulae for those are:

$$\phi(3,1,w) = \frac{1}{3}\{\phi(w) + 2^{s-1}\} \text{ for } w = 1 \pmod 3$$

and

$$\phi(3,1,w) = \frac{1}{6}\{\phi(w) - 2^{s-1}\} \text{ for } w = -1 \pmod 3$$

$$\phi(6,1,w) = \frac{1}{3}\{\phi(w) + 2^{s+1}\} \text{ for } w = 1 \pmod 3$$

and

$$\phi(6, 1, w) = \frac{1}{6}\left\{\phi(w) - 2^{s+1}\right\} \text{ for } w = -1 \pmod 3$$

3. PI3(n) and PI6(n) are the much faster computations of the (brute-force) objects PI(n, 3, 1) and PI(n, 6, 1).

It will be observed they put into effect the $M = 3$ and 6 cases of the closed forms Theory set out in our paper.

Another worksheet exhibits the (absolutely *fundamental*) closed forms for those cases. "Fundamental", not just from a theory point of view, but for computation purposes besides.

Procedures (with some illustrative computational examples)

```
> with(numtheory):

### 01:

PI := proc(n, M, i) local k, r; r := 1:
      for k from floor(((i-1)*(n-1)/M + 1)) to i*(n-1)/M do
        if igcd(n, k) = 1 then r := mods(r*k, n); fi; od; r; end:

### 02:

PRFAC := proc(le, la, p, alpha) local r, k, MOD; r := le; MOD := p^alpha;

         for k from (le+1) to la do r := mods(r*k, MOD) od; r; end:

### 03:

residues := proc(n, M)

  [seq(mods(op(i, factorset(n)), M), i = 1..nops(factorset(n)))] end:

### 04:

the_ones := proc(n, M) local L, p; L := []: for p in factorset(n) do

    if p mod M = 1 then L := [op(L), p] fi od; L; end:

the_minus_ones := proc(n, M) local L, p; L := []: for p in factorset(n) do

    if mods(p, M) = -1 then L := [op(L), p] fi od; L; end:

### 05:

Pow := proc(n, p) local t, a; t := n: a := 0: while t mod p = 0 do

            t := t/p: a := a+1: od: a; end:

### 06:

PHI3 := proc(w) local r; r := nops(factorset(w)):

      if w mod 3 = 1 then (phi(w) + 2^(r-1))/3

    elif mods(w, 3) = -1 then (phi(w) - 2^(r-1))/3 fi; end:

### 07:

PI3 := proc(n) local p, a, Gf, w, s, signs, PHIw3, Sw, Q, EF, Rpa, Rw, R;

        p := op(the_ones(n, 3)):     ### This gives 'p'
        a := Pow(n, p):              ### This gives 'a', i.e. 'alpha'
```

```
      if a = 1 then Gf := PRFAC(1, (p-1)/3, p, 1)
    elif a > 1 then Gf := PI(p^a, 3, 1) fi:

       ### This is ((p^a - 1)/3)_p! mod p^a

        w := n/(p^a):                ### This is 'w'
        s := nops(factorset(w)):     ### This is 's'
    signs := (-1)^(s-1):  ### This is '1' at ODD 's', and '-1' at EVEN 's'
    PHIw3 := PHI3(w):

       Sw := factorset(w):    ### This is the set of all ('s' of) the 'q'
        Q := mul(q, q = Sw):   ### This is the product of all ('s' of) the 'q'
       EF := mods(Q&^(phi(p^a)/3), p^a): ### the Euler-Fermat element.

      Rpa := mods(EF^signs * Gf&^(2^s), p^a):
             ### Note the SIGN element in the EF term
       Rw := mods(1/p&^PHIw3, w):
             ### There is NO SIGN element here at Gauss 3

    if w = 2 then

         R := mods(chrem([-1/Rpa, Rw], [p^a, w]), n): ### Note the '-'

  elif w <> 2 and mods(w, 3) = -1 then

         R := mods(chrem([1/Rpa, Rw], [p^a, w]), n):  ### Note the 1/Rpa

  elif mods(w, 3) = 1 then

         R := mods(chrem([Rpa, Rw], [p^a, w]), n):

    fi; R; end:

### 08:

PHI6 := proc(w) local r; r := nops(factorset(w)):

      if w mod 6 = 1 then (phi(w) + 2^(r+1))/6
     elif mods(w, 6) = -1 then (phi(w) - 2^(r+1))/6 fi; end:

### 09:

PI6 := proc(n) local p, a, Gf, w, s, signs, PHIw6,

       Sw, Q, PARI, EF1, q1, sign1, EF, Rpa, Rw, R;

      p := op(the_ones(n, 6)):  ### This gives 'p'
      a := Pow(n, p):           ### This gives 'a', i.e. 'alpha'

     if a = 1 then Gf := PRFAC(1, (p-1)/6, p, 1)
    elif a > 1 then Gf := PI(p^a, 6, 1) fi:

      ### This is ((p^a - 1)/6)_p! mod p^a

      w := n/(p^a):                              ### This is 'w'

   PARI := proc(w)
                  if mods(w, 6) = 1 then 1
                elif mods(w, 6) = -1 then -1 fi end:

      s := nops(factorset(w)):   ### This is 's'
  signs := (-1)^s:      ### This is '-1' at ODD 's', and '1' at EVEN 's'
  PHIw6 := PHI6(w):

     Sw := factorset(w):  ### This is the set of all ('s' of) the 'q'
      Q := mul(q, q = Sw): ### This is the product of all ('s' of) the 'q'

    ### We need TWO EULER-FERMATS (EF1 and EF)
    ### to distinguish between s = 1 and s > 1:

    if s = 1 then

    q1 := op(1, factorset(w)):     ### This is 'q[1]'
 sign1 := (-1)^((p+q1)/6):         ### the sign element in the closed form
   EF1 := mods(q1&^(phi(p^a)/6), p^a): ### the Euler-Fermat element for q[1].
                                        ### NOTE the 6th-root
```

```
   Rpa := mods(sign1 * EF1 * Gf^2, p^a):
    Rw := mods((-1)^((p-1)/6)/p&^PHIw6, w):
          ### Note the EXTRA SIGN element here at Gauss 6

     R := mods(chrem([Rpa^PARI(w), Rw], [p^a, w]), n):

        ### Note the 1/Rpa, as with Gauss 4 closed forms

   elif s > 1 then

   EF := mods(Q&^(phi(p^a)/3), p^a):  ### Euler-Fermat element for s > 1.
                                        ### Note the 3rd-root
  Rpa := mods(EF^signs*Gf^(2^s), p^a): ### Note the SIGN element in the EF term
   Rw := mods(1/p&^PHIw6, w):   ### There is NO SIGN element here at s > 1

    R := mods(chrem([Rpa^PARI(w), Rw], [p^a, w]), n):

    fi; R; end:
```

```
>      p := 31:
   alpha := 2:
      q1 := 5:
   beta1 := 2:
      q2 := 11:
   beta2 := 2:
       n := p^alpha * q1^beta1 * q2^beta2:

     [n mod 3, n mod 6]; ifactor(n);

  M := 3; [PI(n, M, 1), PI3(n)];
  M := 6; [PI(n, M, 1), PI6(n)];
```

$$[1, 1]$$

$$(5)^2\,(11)^2\,(31)^2$$

$$M := 3$$

$$[-1377994, -1377994]$$

$$M := 6$$

$$[-1265964, -1265964] \quad (1.1)$$

```
>      p := 31:
   alpha := 2:
      q1 := 5:
   beta1 := 1:  ### Note the changed value
      q2 := 11:
   beta2 := 2:

       n := p^alpha * q1^beta1 * q2^beta2:

     [mods(n, 3), mods(n, 6)]; ifactor(n);

  M := 3; [PI(n, M, 1), PI3(n)];
  M := 6; [PI(n, M, 1), PI6(n)];
```

$$[-1, -1]$$

$$(5)\,(11)^2\,(31)^2$$

$$M := 3$$

$$[-194134, -194134]$$

$$M := 6$$

$$[175046, 175046] \quad (1.2)$$

Table #1 showing the 26 solutions of $\left(\frac{1}{3}\left\{n-1\right\}\right)_n! = 1 \pmod{n}$ **with** $n = -1 \pmod{3}$ **(i.e.** $w = -1 \pmod{3}$**)**

```
> L3_minus_1 := []: for n from 2 by 3 to 10^6 do if
   nops(factorset(n)) > 1 and nops(the_ones(n, 3)) = 1 and PI(n, 3, 1) = 1
           then L3_minus_1 := [op(L3_minus_1), n] fi od:
  for n in L3_minus_1 do
                                    p||n := op(1, the_ones(n, 3)):
                                alpha||n := Pow(n, p||n):
                                    w||n := n/(p||n^alpha||n):
                                   R3||n := PI3(n):
  od: n := 'n': print(``);
  print(array([['n',``,'n_mods_3',``,'p^alpha',``,'w',``,({n-1}/3)[n]!],
    seq([n,``,[mods(n, 3)],``,ifactor(p||n^alpha||n),``,ifactor(w||n),``,R3||n],
  n = L3_minus_1)])): print(``);
  lprint(`Observe the final column is re-computed with CLOSED FORMS PI3 procedure.`);
```

n	n_mods_3	p^{α}	w	$\left(\frac{1}{3}\{n-1\}\right)_n!$
26	[−1]	(13)	(2)	1
305	[−1]	(61)	(5)	1
338	[−1]	$(13)^2$	(2)	1
9755	[−1]	(1951)	(5)	1
60707	[−1]	(3571)	(17)	1
70673	[−1]	(2437)	(29)	1
95990	[−1]	(331)	$(2)\,(5)\,(29)$	1
101651	[−1]	(9241)	(11)	1
165380	[−1]	(8269)	$(2)^2\,(5)$	1
167690	[−1]	(409)	$(2)\,(5)\,(41)$	1
184820	[−1]	(9241)	$(2)^2\,(5)$	1
211178	[−1]	(331)	$(2)\,(11)\,(29)$	1
224204	[−1]	(2437)	$(2)^2\,(23)$	1
232373	[−1]	(13669)	(17)	1
274730	[−1]	(331)	$(2)\,(5)\,(83)$	1
297743	[−1]	(10267)	(29)	1
383960	[−1]	(331)	$(2)^3\,(5)\,(29)$	1
516644	[−1]	(2437)	$(2)^2\,(53)$	1
604406	[−1]	(331)	$(2)\,(11)\,(83)$	1
605753	[−1]	(10267)	(59)	1
633455	[−1]	(126691)	(5)	1
670760	[−1]	(409)	$(2)^3\,(5)\,(41)$	1
739280	[−1]	(9241)	$(2)^4\,(5)$	1
749390	[−1]	(547)	$(2)\,(5)\,(137)$	1
844712	[−1]	(331)	$(2)^3\,(11)\,(29)$	1
950249	[−1]	(55897)	(17)	1

`Observe the final column is re-computed with CLOSED FORMS PI3 procedure.` **(2.1)**

Table #2 showing the 22 solutions of $\left(\frac{1}{3}\{n-1\}\right)_n! = 1 \pmod{n}$ with $n = 1 \pmod{3}$ (i.e. $w = 1 \pmod{3}$)

```
> L3_plus_1 := []:
  for n from 4 by 3 to 10^6 do if
   nops(factorset(n)) > 1 and nops(the_ones(n, 3)) = 1 and PI(n, 3, 1) = 1
           then L3_plus_1 := [op(L3_plus_1), n] fi od:
  for n in L3_plus_1 do
                                p||n := op(1, the_ones(n, 3)):
                            alpha||n := Pow(n, p||n):
                                w||n := n/(p||n^alpha||n):
                               R3||n := PI3(n): ### the CLOSED FORMS COMPUTATION
  od: n := 'n':
  print(array([['n',``,'n_mods_3',``,'p^alpha',``,'w',``,({n-1}/3)[n]!],
      seq([n,``,[mods(n, 3)],``,ifactor(p||n^alpha||n),``,ifactor(w||n),``,R3||n],
  n = L3_plus_1)])): print(``);
  lprint(`Observe the final column is re-computed with CLOSED FORMS PI3 procedure.`);
```

n	n_mods_3	p^{α}	w	$\left(\frac{1}{3}\{n-1\}\right)_n!$
244	[1]	(61)	$(2)^2$	1
18205	[1]	(331)	$(5)(11)$	1
33076	[1]	(8269)	$(2)^2$	1
48775	[1]	(1951)	$(5)^2$	1
82690	[1]	(8269)	$(2)(5)$	1
92410	[1]	(9241)	$(2)(5)$	1
112102	[1]	(2437)	$(2)(23)$	1
191980	[1]	(331)	$(2)^2(5)(29)$	1
258322	[1]	(2437)	$(2)(53)$	1
330760	[1]	(8269)	$(2)^3(5)$	1
335380	[1]	(409)	$(2)^2(5)(41)$	1
369640	[1]	(9241)	$(2)^3(5)$	1
422356	[1]	(331)	$(2)^2(11)(29)$	1
448408	[1]	(2437)	$(2)^3(23)$	1
516934	[1]	(23497)	$(2)(11)$	1
549460	[1]	(331)	$(2)^2(5)(83)$	1
583444	[1]	(145861)	$(2)^2$	1
609190	[1]	(60919)	$(2)(5)$	1
767920	[1]	(331)	$(2)^4(5)(29)$	1
808084	[1]	(202021)	$(2)^2$	1
876514	[1]	(8269)	$(2)(53)$	1
924010	[1]	(92401)	$(2)(5)$	1

`Observe the final column is re-computed with CLOSED FORMS PI3 procedure.`

(3.1)

Table #3 showing the 5 solutions of $\left(\frac{1}{6}\{n-1\}\right)_n! = 1 \pmod{n}$ with $n = -1 \pmod{6}$ (i.e. $w = -1 \pmod{6}$)

```
> L6_minus_1 := []: for n from 7 by 6 to 10^6 do if
   nops(factorset(n)) > 1 and nops(the_ones(n, 6)) = 1 and PI(n, 6, 1) = 1
           then L6_minus_1 := [op(L6_minus_1), n] fi od:
  for n in L6_minus_1 do
                              p||n := op(1, the_ones(n, 6)):
                          alpha||n := Pow(n, p||n):
                              w||n := n/(p||n^alpha||n):
                             R6||n := PI6(n):
  od: n := 'n':
  print(array([['n',``,'n_mods_6',``,'p^alpha',``,'w',``,({n-1}/6)[n]!],
     seq([n,``,[mods(n, 6)],``,ifactor(p||n^alpha||n),``,ifactor(w||n),``,R6||n],
  n = L6_minus_1)])): print(``);
  lprint(`Observe the final column is re-computed with the CLOSED FORMS PI6 procedure.
  `);
```

n	n_mods_6	p^{α}	w	$\left(\frac{1}{6}\{n-1\}\right)_n!$
309485	$[-1]$	(331)	$(5)\,(11)\,(17)$	1
510605	$[-1]$	(102121)	(5)	1
527945	$[-1]$	(331)	$(5)\,(11)\,(29)$	1
729305	$[-1]$	(145861)	(5)	1
746405	$[-1]$	(331)	$(5)\,(11)\,(41)$	1

`Observe the final column is re-computed with the CLOSED FORMS PI6 procedure.`

(4.1)

Table #4 showing the 9 solutions of $\left(\frac{1}{6}\{n-1\}\right)_n! = 1 \pmod{n}$ with $n = 1 \pmod{6}$ (i.e. $w = 1 \pmod{6}$)

```
> L6_plus_1 := []: for n from 7 by 6 to 10^6 do if
```

```
 nops(factorset(n)) > 1 and nops(the_ones(n, 6)) = 1 and PI(n, 6, 1) = 1
         then L6_plus_1 := [op(L6_plus_1), n] fi od:
for n in L6_plus_1 do
                              p||n := op(1, the_ones(n, 6)):
                          alpha||n := Pow(n, p||n):
                              w||n := n/(p||n^alpha||n):
                             R6||n := PI6(n):
od: n := 'n':
print(array([['n',``,'n_mods_6',``,'p^alpha',``,'w',``,({n-1}/6)[n]!],
   seq([n,``,[mods(n, 6)],``,ifactor(p||n^alpha||n),``,ifactor(w||n),``,R6||n],
n = L6_plus_1)])): print(``);
lprint(`Observe the final column is re-computed with the CLOSED FORMS PI6 procedure.
`);
```

n	n_mods_6	p^{α}	w	$\left(\frac{1}{6}\{n-1\}\right)_n!$
1105	[1]	(13)	(5) (17)	1
14365	[1]	$(13)^2$	(5) (17)	1
34765	[1]	(409)	(5) (17)	1
303535	[1]	(3571)	(5) (17)	1
353365	[1]	(2437)	(5) (29)	1
508255	[1]	(9241)	(5) (11)	1
796717	[1]	(331)	(29) (83)	1
839185	[1]	(3571)	(5) (47)	1
872695	[1]	(10267)	(5) (17)	1

`Observe the final column is re-computed with the CLOSED FORMS PI6 procedure.`

(5.1)